Polynômes et Fractions rationnelles

```
> restart:
```

Les opérations de base sur les polynômes sont données par collect,sort,expand qui permettent d'écrire convenablement un polynôme,

ou une expression apparentée.

```
> f := (x^2+x+2): g := (x^2+x*y +y^4):
> f^3;expand(f^3);
```

$$(x^2 + x + 2)^3$$

$$x^6 + 3\,x^5 + 9\,x^4 + 13\,x^3 + 18\,x^2 + 12\,x + 8$$

```
> h := expand(f*g);
```

$$h := x^4 + x^3\,y + x^2\,y^4 + x^3 + x^2\,y + x\,y^4 + 2\,x^2 + 2\,x\,y + 2\,y^4$$

```
> sort(h,x);sort(h,y);
```

$$x^4 + y\,x^3 + x^3 + y^4\,x^2 + y\,x^2 + 2\,x^2 + y^4\,x + 2\,y\,x + 2\,y^4$$

$$2\,y^4 + x\,y^4 + x^2\,y^4 + x^3\,y + x^2\,y + 2\,x\,y + 2\,x^2 + x^3 + x^4$$

```
> collect(h,x);collect(h,y);
```

$$x^4 + (y + 1)\,x^3 + (y^4 + y + 2)\,x^2 + (y^4 + 2\,y)\,x + 2\,y^4$$

$$(x^2 + x + 2)\,y^4 + (2\,x + x^2 + x^3)\,y + x^4 + 2\,x^2 + x^3$$

indets(f), degre(f,x), coeff(f,x,k) retournent les indéterminées, le degré, le coefficient de x^k dans le polynôme.

```
> indets(g);
```

$$\{x,\, y\}$$

```
> coeff(h,y,4);
```

> $x^2 + x + 2$

$$x^2 + x + 2$$

factor factorise le polynôme si possible.

```
> factor(x^8-1);
```

$$(x - 1)\,(x + 1)\,(x^2 + 1)\,(x^4 + 1)$$

On peut effectuer une division euclidienne, factoriser, calculer le résultant et/ou le discriminant : toujours préciser le nom de la variable.

```
> f := x^4+4*x^3-3*x+7: g:= x^2+2*x +3:
> q := quo(f,g,x);
```

$$q := x^2 + 2\,x - 7$$

```
> r := rem(f,g,x);
```

$$r := 28 + 5\,x$$

```
> expand(q*g+r);
```

$$x^4 + 4\,x^3 - 3\,x + 7$$

```
> resultant(f,g,x);
```

$$579$$

```
> f := x^4-1:g:=x^2+2*x-3:resultant(f,g,x);
```

$$0$$

Il y a aussi les fonctions gcd et lcm pour le pgcd et le ppcm.
La fonction factor retourne une factorisation d'un polynôme à coefficients entiers en polynômes *irréductibles sur Q*.

```
> f :=(x^4-4)*(x^3+8): factor(f);
```

$$(x+2)\,(x^2+2)\,(x^2-2)\,(x^2-2\,x+4)$$

Plus généralement elle décompose dans l'extension algébrique de Q contenant les coefficients du polynôme.

```
> g:=(x^2-2)*(x-sqrt(2)): factor(g);
```

$$(x+\sqrt{2})\,(x-\sqrt{2})^2$$

La fonction roots permet d'obtenir les racines dans la même extension algébrique, alors que solve les donne toutes.

```
> roots(f);solve(f);
```

$$[[-2, 1]]$$

$$\sqrt{2},\ -\sqrt{2},\ I\,\sqrt{2},\ -I\,\sqrt{2},\ -2,\ 1 - I\,\sqrt{3},\ 1 + I\,\sqrt{3}$$

Pour forcer la décomposition dans une certaine extension on utilise la variante

```
> factor(f,sqrt(2));
```

$$-(x+2)\,(x^2+2)\,(-x+\sqrt{2})\,(x+\sqrt{2})\,(x^2-2\,x+4)$$

```
> factor(f,{sqrt(2),I});
```

$$(x+I\,\sqrt{2})\,(x+2)\,(-x+\sqrt{2})\,(-x+I\,\sqrt{2})\,(x+\sqrt{2})\,(x^2-2\,x+4)$$

```
> factor(f,{sqrt(3),I});
```

$$-(x+2)\,(x^2+2)\,(x^2-2)\,(-x+1+I\,\sqrt{3})\,(x-1+I\,\sqrt{3})$$

La fonction RootOf permet de désigner une racine d'un polynôme, associée avec l'intruction alias c'est un outil puissant.

```
> f := x^6-2*x+1: solve(f);
```

$$1, \mathrm{RootOf}(\_Z^5 + \_Z^4 + \_Z^3 + \_Z^2 + \_Z - 1)$$

```
> g := x^3-2;
```

$$g := x^3 - 2$$

```
> a:= RootOf(g,x);
```

$$a := \mathrm{RootOf}(\_Z^3 - 2)$$

```
> factor(x^6-4,a);
```

$$(x^2 - \mathrm{RootOf}(\_Z^3 - 2)\,x + \mathrm{RootOf}(\_Z^3 - 2)^2)\,(x^2 + \mathrm{RootOf}(\_Z^3 - 2)\,x + \mathrm{RootOf}(\_Z^3 - 2)^2)\,(x + \mathrm{RootOf}(\_Z^3 - 2))\,(x - \mathrm{RootOf}(\_Z^3 - 2))$$

Ici on a fait une affectation a :=RootOf mais on peut utiliser un alias, l'avantage est que Maple substitue aussi dans le résultat. A chaque nouvel alias Maple

retourne la liste des autres alias, par défaut le seul connu est I=RootOf(_Z^2-1).

```
> alias(b=RootOf(g,x));
```

$$I,\, b$$

```
> factor(x^6-4,b);
```

$$(x^2 - b\,x + b^2)\,(x^2 + b\,x + b^2)\,(x + b)\,(x - b)$$

```
> f:= x^3-3*x: g:= diff(f,x):
> alias(c=RootOf(g));
```

$$I,\, b,\, c$$

```
> subs(x=c,f);simplify(");
```

$$c^3 - 3\,c$$

$$-2\,c$$

```
> evalf(c);
```

$$-1.000000000$$

RootOf désigne une racine du polynôme, mais on peut aussi considérer l'ensemble des racines grâce à l'instruction allvalues .

```
> bb := allvalues(b); allvalues(c);
```

$$bb := 2^{1/3},\ -\frac{1}{2}\,2^{1/3} + \frac{1}{2}\,I\,\sqrt{3}\,2^{1/3},\ -\frac{1}{2}\,2^{1/3} - \frac{1}{2}\,I\,\sqrt{3}\,2^{1/3}$$

$$1,\ -1$$

```
> sort([bb]);
```

$$[2^{1/3},\ -\frac{1}{2}\,2^{1/3} - \frac{1}{2}\,I\,\sqrt{3}\,2^{1/3},\ -\frac{1}{2}\,2^{1/3} + \frac{1}{2}\,I\,\sqrt{3}\,2^{1/3}]$$

```
> allvalues(b+c);
```

$$2^{1/3}+1,\ 2^{1/3}-1,\ -\frac{1}{2}\,2^{1/3}+\frac{1}{2}\,I\,\sqrt{3}\,2^{1/3}+1,\ -\frac{1}{2}\,2^{1/3}+\frac{1}{2}\,I\,\sqrt{3}\,2^{1/3}-1,\ -\frac{1}{2}\,2^{1/3}-\frac{1}{2}\,I\,\sqrt{3}\,2^{1/3}+1,\ -\frac{1}{2}\,2^{1/3}-\frac{1}{2}\,I\,\sqrt{3}\,2^{1/3}-1$$

On peut convertir les RootOf en radicaux et vice-versa, mais il faut éviter le mélange des deux ; attention s'il y a des imaginaires ils sont définis par des
RootOf.
Exemple avec des fonctions symétriques des racines

```
> f := x^3+x+1: sum(1/(x+1),x=RootOf(f));
```

$$4$$

Enfin on peut travailler dans Z/pZ (avec p premier !) en utilisant l'instruction mod est les formes inertes Factor, Quo, Rem ....

```
> f := x^4+x^2+1: Factor(f) mod 13;
```

$$(x+10)\,(x+9)\,(x+3)\,(x+4)$$

```
> expand(");
```

$$x^4 + 26\,x^3 + 235\,x^2 + 858\,x + 1080$$

```
> " mod 13;
```

$$x^4 + x^2 + 1$$

Pour les fractions rationelles, on dispose de la forme normal, des expressions numer, denom et toujours factor. Pour la décomposition en éléments
simples on utilise convert associé à parfrac .

```
> f := 1/(x+1) + (2*x+1)/(x^2+x+1);
```

$$f := \frac{1}{x+1} + \frac{2\,x+1}{x^2+x+1}$$

```
> normal(f);
```

$$\frac{3\,x^2+4\,x+2}{(x+1)\,(x^2+x+1)}$$

```
> numer(f);
```

$$3\,x^2+4\,x+2$$

```
> factor( x + x^3/(x^2+2*x+1));
```

$$\frac{x\,(2\,x^2+2\,x+1)}{(x+1)^2}$$

```
> g := (x^2+4)/(x^4-x^2+1):
> factor(g,sqrt(3));
```

$$\frac{x^2+4}{(x^2-\sqrt{3}\,x+1)\,(x^2+\sqrt{3}\,x+1)}$$

```
> convert((x^4+1)/(x^4-1),parfrac,x);
```

$$1+\frac{1}{2}\,\frac{1}{x-1}-\frac{1}{2}\,\frac{1}{x+1}-\frac{1}{x^2+1}$$

```
> convert(g,parfrac,x);
```

$$\frac{x^2+4}{x^4-x^2+1}$$

Si on n'essaye pas d'aider Maple en indiquant une décomposition ...

```
> gd := factor(denom(g),sqrt(3));
```

$$gd := (x^2-\sqrt{3}\,x+1)\,(x^2+\sqrt{3}\,x+1)$$

```
> convert(numer(g)/gd,parfrac,x);
```

$$\frac{1}{2}\,\frac{-4+\sqrt{3}\,x}{-x^2+\sqrt{3}\,x-1}+\frac{1}{2}\,\frac{4+\sqrt{3}\,x}{x^2+\sqrt{3}\,x+1}$$

```
> cgd := factor(denom(g),{I,sqrt(3)});
```

$$cgd := \frac{1}{16}\,(2\,x+\sqrt{3}-I)\,(2\,x-\sqrt{3}-I)\,(2\,x+\sqrt{3}+I)\,(2\,x-\sqrt{3}+I)$$

```
> convert(numer(g)/cgd,parfrac,x);
```

$$\frac{1}{2}\,\frac{\sqrt{3}-5\,I}{2\,x+\sqrt{3}-I}+\frac{1}{2}\,\frac{\sqrt{3}+5\,I}{-2\,x+\sqrt{3}+I}+\frac{1}{2}\,\frac{\sqrt{3}+5\,I}{2\,x+\sqrt{3}+I}+\frac{1}{2}\,\frac{\sqrt{3}-5\,I}{-2\,x+\sqrt{3}-I}$$

Polynôme d'interpolation, fonctions par morceaux.

Pour touver le polynôme d'interpolation prenant les valeurs y_i aux points x_i, on fait une liste des valeurs de x, une liste des valeurs de y et on

appelle la fonction interp

```
> xx := [0,1,2,4]: yy := [1,2,9,65]: p:=
> interp(xx,yy,x);
```

$$p := x^3 + 1$$

Pour définir une fonction par morceaux, on peut utiliser une procedure ou l'instruction piecewise, l'avantage de cette dernière solution est qu'on

peut lui appliquer certains opérateurs.

```
> restart:
> f := proc(x)  if (abs(x) < 1) then RETURN(1);
> else RETURN(0); fi; end;
```

$$f := \textbf{proc}(x)\,\textbf{if}\;\mathrm{abs}(x) < 1\;\textbf{then}\;\mathrm{RETURN}(1)\;\textbf{else}\;\mathrm{RETURN}(0)\;\textbf{fi}\;\textbf{end}$$

```
> plot(f,-2..2);
```




```
> int(f(x),x=-2..2);
```

```
Error, (in f) cannot evaluate boolean
```

```
> ff := piecewise(abs(x)<1,1);
```

$$ff := \begin{cases} 1 & |x| < 1 \\ 0 & \textit{otherwise} \end{cases}$$

```
> gg := x -> int(ff(s),s=-1..x);
```

$$gg := x \to \int_{-1}^{x} \mathrm{ff}(s)\, ds$$

```
> plot(gg(x),x=-2..2);
```